
Subject: Re: Treating an object as a structure

Posted by [Michael Galloy](#) on Wed, 04 Mar 2009 23:45:16 GMT

[View Forum Message](#) <> [Reply to Message](#)

Ilo wrote:

```
> Ok, thanks ...
>
> With this discussion, all of a sudden I wondered how ITT people
> implement the ALL keyword in GetProperty method of the objects ?
> Because maybe the easiest way to get all the content of the object is
> having the ALL keyword in GetProperty method. ( I know I said I don't
> wanna use this method )
>
> I don't know how can I implement this method. Maybe I will try
> something like this:
>
> PRO object::GetProperty, A=a, B=b, C=c, D=d, E=e, F=f, G=g, ALL=all
>
> ....
>
> IF ARG_PRESENT(all) THEN BEGIN
>   all=CREATE_STRUCT
>   ('A',self.a,'B',self.b,'C',self.c,'D',self.d,'E',self.e,'F', self.f,'G',self.g)
>   ENDIF
> END
>
> Is it correct or there are a better way to do that ?
```

Just use STRUCT_ASSIGN:

```
pro mg_instance_test::getProperty, all=all
  compile_opt strictarr

  if (arg_present(all)) then begin
    all = create_struct(name=obj_class(self))
    struct_assign, self, all
  endif
end
```

```
function mg_instance_test::init
  compile_opt strictarr

  self.a = 3.0
  self.b = 'Hello'
  self.c = 123L

  return, 1
```

end

```
pro mg_instance_test__define
  compile_opt strictarr

  define = { mg_instance_test, a: 0.0, b: "", c: 0L }
end
```

Run the example like:

```
IDL> o = obj_new('mg_instance_test')
IDL> o->getProperty, all=all
IDL> help, all, /structures
IDL> obj_destroy, o
```

Mike

--

www.michaelgalloy.com
Associate Research Scientist
Tech-X Corporation
